
Na uzorku od 737 mu{karaca, starih od 19 do 27 godina, izmerene su, na na~in koji minimizira pogre{ku merenja, 23 antropometrijske varijable kojima su procenjeni faktori longitudinalne i transverzalne dimenzionalnosti skeleta, mi{i}ne mase i masnog tkiva. Morfolo{ki tipovi odre|eni su neuronskom mre`om SIMTAX koja klasifikuje ispitanike u prostoru standardizovanih image varijabli iterativnom primenom Lebartovog troslojnog perceptrona. Inicijalnu klasifikaciju program kojim je implmentirana ova mre`a formira na osnovu polo`aja objekata na obodu hiperelipsoida definisanog Orthoblique transformacijama komponenata ~iji je broj, a time i broj taksona, definisan brojem spektralnih vrednosti ve}ih od ta~ke pregiba njihove distribucije. Dobijena su tri taksona sa efikasno{}u klasifikacije od 0.991 u image i 0.986 u realnom prostoru. Prvi takson, u kome je bilo 35% ispitanika, identifikovan je kao astenomorfija, drugi, u kome je bilo 29% ispitanika, kao stenomorfija a tre}i, u kome je bilo 36% ispitanika, kao piknomorfija. Taksoni su identifikovani na osnovu rezultata Fiherove metode diskriminativne analize, tako|e provedene u standardizovanom image prostoru. U prvom taksonu na{li su se ispitanici sa vrlo malom koli~inom aktivne mi{i}ne mase, malom koli~inom masnog tikiva i malim transverzalnim, ali prose~nim ili iznadprose~nim longitudinalnim dimenzijama skeleta. U drugom taksonu bili su ispitanici sa visokim vrednostima na svim antropometrijskim varijablama, a naro~ito na merama mi{i}ne mase. U tre}em su taksonu bili ispitanici sa malim i longitudinalnim i transverzalnim dimenzijama skeleta, malom koli~inom masnog tikva i ispodprose~nom koli~inom mi{i}ne mase. Prema tome, ni Sheldonovi ni Kretchmerovi tipovi ne postoje kada se radi o mu{karcima koji se nalaze u stacionarnoj fazi morfolo{kog razvoja, a od pet tipova koje je, analizom razlivenih taksona, prona{la A. Ho{ek, identifikovani su samo tipovi M, K i R.

KLJU^NE RE^I

morfolo{ki tipovi / taksonomska analiza / neuronske mre`e

1. UVOD

Problem objektivnog odre|ivanja morfolo{kih tipova jo{ uvek je jedan od najkontroverznijih problema u biolo{koj antropologiji. U jednom nedavno objavljenom radu A. Ho{ek tvrdi da kako velika ve}ina antropolo{kih, dakle morfolo{kih, fiziolo{kih, psiholo{kih i sociolo{kih karakteristika ima, aproksimativno, normalnu distribuciju, i kako koherentni skupovi tih karakteristika imaju, gotovo uvek, multivarijatnu normalnu distribuciju, metode za odre|ivanje antropolo{kih tipova pod modelom distinktnih taksona nisu primerene distributivnim svojstvima tih karakterisrika. Zbog toga smatra da je za odre|ivanje antropolo{kih tipova razumnije primeniti neku metodu koja pripada skupu metoda za analizu razlivenih taksona, razmatra jednu klasu tavkvih algoritama, koji pripadaju skupu metoda za analizu polarnih taksona, i predla`e neka re{enja problema izbora adekvatne metrike, odre|ivanja broja taksonomskih dimenzija, i izbora najpogodnije parsimonijske funkcije (Ho{ek, 2002).


Me|utim, nije sasvim sigurno da se metodom formiranja distinktinih taksona ne mogu dobiti smisleni rezultati jer, iako su, ako se izuzmu ko`ni nabori, marginalne distribucije svih antropometrijskih varijabli aproksimativno normalne, to ne zna~i nu`no da je i multivarijatna distibucija tih varijabli zaista normalna. Zbog toga }e u ovom radu biti prikazani rezultati taksonomske analize izvedene jednom neuronskom mre`om
 koja klasifikuje entitete opisane antropometrijskim varijablama pretvorenim u standardizovani image oblik, dakle u oblik koji je najnepovoljniji sa taksonomske ta~ke gledi{ta. Jer, ako se i u ovom prostoru mogu otkriti smisleni distinkni morfolo{ki taksoni, to zna~i da morfolo{ki tipovi zaista postoje, i da je problem samo da se za njihovu detekciju primeni neki adekvatni metodolo{ki postupak.

2. METODE


Slu~ajni uzorak od 737 ispitanika, izabran iz populacije mu{karaca starih od 19 do 27 godina, izmeren je sa 23 antropometrijske mere na na~in koji minimizira pogre{ku merenja (Stojanovi}, Solari}, Momirovi} i Vukosavljevi}, 1975). Masa tela, skeletalne i cirkularne dimenzije merene su po tri puta, a ko`ni nabori po {est puta. Iz ovih mera, kao prve glavne komponente u prostoru sa univerzalnom metrikom, derivirane su ove varijable:

KODNI NAZIV
VARIJABLA

TEZINA
Masa tela

VISINA
Visina

DUZINO
Du`ina nogu

BIAKRO
Biakromijalni raspom

BIKRIS
Bikristalni raspon

NANADL
Nabor na nadlaktici

NANALE
Nabor na le|ima

NAPAZU
Nabor na pazuhu

OPNADL
Obim nadlaktice

OPPODL
Obim podlaktice

OPNATK
Obim nadkolenice

OPPOTK
Obim podkolenice

DUZISA
Du`ina {ake

SIRISA
[irina {ake

NATRBU
Nabor na trbuhu

NAPOTK
Nabor na podkolenici

OPGRUD
Obim grudi

DIRUZG
Dijametar ru~nog zgloba

DILAKT
Dijametar lakta

DIKOLJ
Dijametar kolena

DUZIST
Du`ina stopala

SISTOP
[irina stopala

DUZIRU
Dijametar ru~nog zgloba


Rezultati su analizirani taksonomskom neuronskom mre`om koja nosi kodni naziv SIMTAX koja klasifikuje ispitanike u prostoru standardizovanih image varijabli iterativnom primenom Lebartovog troslojnog perceptrona. Inicijalnu klasifikaciju program kojim je implmentirana ova mre`a formira na osnovu polo`aja objekata na obodu hiperelipsoida definisanog Orthoblique transformacijama komponenata standardizovanih image varijabli ~iji je broj, a time i broj taksona, definisan brojem spektralnih vrednosti ve}ih od ta~ke pregiba njihove distribucije. Na~in na koji to program izvodi vidi se iz simboli~kog koda programa, koji je napisan u Matrix jeziku tako da se mo`e izvesti u standardnom SPSS okru`enju.

preserve

set printback=off

set decimal=dot

*---------------------------------------------------------------------------

*                             SIMTAX

*     EMULATION OF TAXONOMIC NEURAL NETWORK WITH A HIDDEN LAYER

*       FOR STANDARDIZED IMAGE OF INPUT AND BINARY OUTPUT DATA

*        WITH AUTOMATIC COMPUTATION OF INITIAL CLASSIFICATION

*

*                           Version 1.0.

*                            9.9.2002

*

* SIMTAX can be run with the following statements:

* INCLUDE 'SIMTAX.SPS'.

* SIMTAX VARS=input variables names.

*

* Algorithm is described in

* Momirovic, K. (2002):

* Automatic classification by neural networks in image space.

* Technical report, Institute of criminological and sociological research.

*

* Warning 1: Algorithm accept no necessary initially standardized input data.

* Warning 2: Program print final classification of objects and save it as a

*            binary matrix in file ftax.sav, and as a nominal variable in 

*            file nomtax.sav.

*

*---------------------------------------------------------------------------

define simtax (vars=!charend('/'))

set mxloops=9999

*---------------------------------------------------------------------------

* Saving the original file.

*---------------------------------------------------------------------------

save outfile='tmp__fds.sav'

*---------------------------------------------------------------------------

* Activation of matrix language and preliminary operations.

*---------------------------------------------------------------------------

matrix

print /title '           *** S I M T A X ***' /space 1

print /title 'AUTOMATIC CLASSIFICATION BY NEURAL NETWORK'/space 0

print /title '      IN STANDARDIZED IMAGE SPACE'/space 0

get x/file=*/variables=!vars/names=nx

compute num=nrow(x)

compute nvr=ncol(x)

print num/format "f8.0"/title 'Number of objects'

print nvr/format "f8.0"/title 'Number of input variables'

compute ent=make(num,1,1)

loop i=1 to num-1

.compute ent(i+1)=ent(i)+1

end loop

compute en=make(num,1,1)

compute mmm=csum(x)&/num

compute mmmm=en*mmm

compute x=x-mmmm

compute c=(t(x)*x)&/num

compute dc=sqrt(ginv(mdiag(diag(c))))

compute x=x*dc

compute rmat=dc*c*dc

compute rmat=(rmat+t(rmat))&/2

compute imat=mdiag(diag(rmat))

compute rinv=inv(rmat)

compute u2=ginv(mdiag(diag(ginv(rmat))))

compute gmat=imat-rinv*u2

compute x=x*gmat

compute gmat=(t(x)*x)&/num

compute dmat=inv(sqrt(mdiag(diag(gmat))))

compute rmat=dmat*gmat*dmat

compute x=x*dmat

release mmm,mmmm,dc,c, dmat,gmat

call eigen(rmat,y,lr)

compute a=(lr(nvr)-lr(1))/(nvr-1)

compute mxd=0

loop k=1 to nvr

compute dd=k*a-lr(k)+lr(1)

do if dd>mxd

compute zok=k-1

compute mxd=dd

end if

end loop

compute ngr=zok

print ngr /title 'Number of taxons under Scree criterion'/space=2

do if ngr=1

.compute ngr=ngr+1

end if

print ngr /title 'Number of accepted taxons'/space=2

compute y=y(:,1:ngr)

compute orttv=0

compute ortnc=0

compute ortstab=1

compute ortcols = ncol(y).

compute ortrows = nrow(y).

compute trans=ident(ortcols,ortcols).

compute ortv = y * trans

loop if (ortstab = 1 and ortnc le 999).

. compute ortsb = csum(ortv &** 4)

. compute ortsa = cssq(ortv)

. compute ortsa = ortsa &* ortsa

. compute ortsb = ortsb &* ortrows

. compute ortsa = (ortsb - ortsa) &/ (ortrows * ortrows)

. compute ortsv = rsum(ortsa)

. compute ortnc=ortnc + 1.

. do if (abs(ortsv-orttv) le 1e-7).

.   compute ortstab=ortstab+1.

. else.

.   compute ortstab=1.

. end if.

. compute orttv=ortsv.

.   compute ortd2=mdiag(diag(t(ortv) * ortv)) &/ ortrows

.   compute ortg=t(y) * (ortv &** 3 - ortv*ortd2)

. call eigen (t(ortg) * ortg, ortgx, ortgl).

. compute ortgl = ginv(sqrt(mdiag(ortgl))).

. compute trans=ortg * (ortgx * ortgl * t(ortgx))

. compute ortv = y * trans

end loop

compute ww=t(trans)*t(y)*rmat*y*trans

compute ww=ginv(sqrt(mdiag(diag(ww))))

compute intax=x*y*trans*ww

compute ww=rmax(intax)

compute vv=make(ngr,1,1)

compute ww=ww*t(vv)

compute intax=intax-ww

compute s=make(num,ngr,0)

loop i=1 to num

loop j=1 to ngr

do if intax(i,j)=0

.compute s(i,j)=1

end if

end loop

end loop

compute s0=s

compute kont=t(s0)*s0

compute dkont=diag(kont)

compute mmat=t(x)*s*ginv(kont)

compute numg={"g1","g2","g3","g4","g5","g6","g7","g8","g9","g10",

      "g11","g12","g13","g14","g15","g16","g17","g18","g19","g20",

      "g21","g22","g23","g24","g25","g26","g27","g28","g29","g30"}

compute numg=numg(1:ngr)

*---------------------------------------------------------------------------

* Multilayer perceptron with a hidden layer.

*---------------------------------------------------------------------------

compute numf={"f1","f2","f3","f4","f5","f6","f7","f8","f9","f10",

      "f11","f12","f13","f14","f15","f16","f17","f18","f19","f20",

      "f21","f22","f23","f24","f25","f26","f27","f28","f29","f30",

      "f31","f32","f33","f34","f35","f36","f37","f38","f39","f40",

      "f41","f42","f43","f44","f45","f46","f47","f48","f49","f50",

      "f51","f52","f53","f54","f55","f56","f57","f58","f59","f60",

      "f61","f62","f63","f64","f65","f66","f67","f68","f69","f70",

      "f71","f72","f73","f74","f75","f76","f77","f78","f79","f80",

      "f81","f82","f83","f84","f85","f86","f87","f88","f89","f90",

      "f91","f92","f93","f94","f95","f96","f97","f98","f99","f100"}

compute rinv=inv(rmat)

compute cmat=t(mmat)*rinv*mmat

compute cmat=(cmat+t(cmat))&/2

call eigen(cmat,bmat,lambda)

compute crit=(csum(lambda))/ngr

compute suma=make(ngr,1,0)

loop i=1 to ngr

do if lambda(i,1) > crit

.  compute suma(i,1)=1

end if

end loop

compute k=csum(suma)

compute llambda=lambda(1:k)

do if k=1

.compute k=k+1

end if

print k /title 'Number of hidden neurons'/space=2

compute lambda=lambda(1:k)

compute bmat=bmat(:,1:k)

compute tbmat=t(bmat)

compute amat=rinv*mmat*bmat

*--------------------------------------------------------------------

* Starting objects classification.

*--------------------------------------------------------------------

compute y=x*amat*tbmat

compute hmat=make(num,ngr,0)

compute kkk=rmax(y)

compute emat=make(ngr,1,1)

compute mmm=kkk*t(emat)

compute atax=y-mmm

loop i=1 to num

loop j=1 to ngr

do if atax(i,j)>=0

.compute hmat(i,j)=1

else

compute hmat(i,j)=0

end if

end loop

end loop

compute trt=csum(hmat)

compute trt=t(trt)

loop i=1 to num

loop j=i to ngr

do if trt(j)>0

.compute hmat(i,j)=hmat(i,j)

end if

end loop

end loop

compute ngr=ncol(hmat)

compute numg=numg(1:ngr)

*--------------------------------------------------------------------

* Efficacy of supervised multilayer perceptron.

*--------------------------------------------------------------------

compute kong=t(s)*hmat

compute prog=diag(kong)

compute vec1=make(ngr,1,1)

compute pprog=vec1-(dkont-prog)&/dkont

*-----------------------------------------------------------------

* Indicator matrix in the starting classification.

*-----------------------------------------------------------------

compute monkey={"objects", numg}

compute smat={ent,hmat}

*---------------------------------------------------------------------------

* Learning.

*---------------------------------------------------------------------------

compute s1=hmat

compute numff={"g1","g2","g3","g4","g5","g6","g7","g8","g9","g10",

      "g11","g12","g13","g14","g15","g16","g17","g18","g19","g20",

      "g21","g22","g23","g24","g25","g26","g27","g28","g29","g30",

      "g31","g32","g33","g34","g35","g36","g37","g38","g39","g40",

      "g41","g42","g43","g44","g45","g46","g47","g48","g49","g50",

      "g51","g52","g53","g54","g55","g56","g57","g58","g59","g60",

      "g61","g62","g63","g64","g65","g66","g67","g68","g69","g70",

      "g71","g72","g73","g74","g75","g76","g77","g78","g79","g80",

      "g81","g82","g83","g84","g85","g86","g87","g88","g89","g90",

      "g91","g92","g93","g94","g95","g96","g97","g98","g99","g100"}

compute hmat2=make(num,ngr,0)

loop iter=1 to 999 if (msum(abs(hmat2-s)) ne 0)

compute hmat2=s

compute mmat=t(x)*s*ginv(t(s)*s)

compute cmat=t(mmat)*rinv*mmat

compute cmat=(cmat+t(cmat))&/2

call eigen(cmat,bmat,lambda)

compute crit=(csum(lambda))/ngr

compute suma=make(ngr,1,0)

loop i=1 to ngr

do if lambda(i,1) > crit

.  compute suma(i,1)=1

end if

end loop

compute k=csum(suma)

do if k=1

.compute k=k+1

end if

compute lambda=lambda(1:k)

compute bmat=bmat(:,1:k)

compute tbmat=t(bmat)

compute amat=rinv*mmat*bmat

compute y=x*amat*tbmat

compute kkk=rmax(y)

compute emat=make(ngr,1,1)

compute mmm=kkk*t(emat)

compute atax=y-mmm

loop i=1 to num

loop j=1 to ngr

do if atax(i,j)>=0

.compute hmat(i,j)=1

else

compute hmat(i,j)=0

end if

end loop

end loop

compute trt=csum(hmat)

compute trt=t(trt)

loop i=1 to num

loop j=i to ngr

do if trt(j)>0

.compute hmat(i,j)=hmat(i,j)

end if

end loop

end loop

compute ngr=ncol(hmat)

compute numg=numg(1:ngr)

compute s=hmat

end loop

print iter/format "f8.0"

          /title 'Number of learning attempts'

*--------------------------------------------------------------------

* Final identification structures.

*--------------------------------------------------------------------

print amat/format "f8.3"

          /title 'Final input to hidden layer axons'

          /rname=nx/cname=numff

print tbmat/format "f8.3"

          /title 'Final hidden layer to output axons'

          /rname=numff/cname=numg

*--------------------------------------------------------------------

* Efficacy of taxonomic neural network.

*--------------------------------------------------------------------

compute kont=t(hmat)*hmat

compute mmat=t(x)*hmat*ginv(kont)

print mmat /format "f8.3"

           /title 'Centroids of final taxons'

           /rname=nx/cname=numg

*---------------------------------------------------------------------------

* Fisherian discriminant analysis

*---------------------------------------------------------------------------

compute rmat=(t(x)*x)&/num

compute dkont=diag(t(hmat)*hmat))

compute rinv=ginv(rmat)

compute beta=rinv*mmat

print beta /format "f8.3"

           /title 'Raw Fisherian discriminant coefficients'

           /rname=nx/cname=numg

compute fvar=t(mmat)*rinv*mmat

compute sigma=sqrt(ginv(mdiag(diag(fvar))))

compute fcor=sigma*fvar*sigma

print fcor/format "f8.3"

          /title 'Correlations of Fisherian discriminant functions'

          /rname=numg/cname=numg

compute facmat=mmat*sigma

print facmat /format "f8.3"

           /title 'Structure of Fisherian discriminant functions'

           /rname=nx/cname=numg

compute patmat=facmat*ginv(fcor)

print patmat /format "f8.3"

           /title 'Pattern of Fisherian discriminant functions'

           /rname=nx/cname=numg

compute gama=beta*sigma

print gama /format "f8.3"

           /title 'Standardized Fisherian discriminant coefficients'

           /rname=nx/cname=numg

*---------------------------------------------------------------------------

* Estimation of taxonomic efficacy of neural network.

*---------------------------------------------------------------------------

compute evec=make(num,1,1)

compute konst=diag(fvar)

compute konst=konst&/2

compute korig=evec*t(konst)

compute ffunc=x*beta-korig

compute hmat1=make(num,ngr,0)

compute konj=rmax(ffunc)

compute emat=make(ngr,1,1)

compute majmun=konj*t(emat)

compute atax=ffunc-majmun

loop i=1 to num

loop j=1 to ngr

do if atax(i,j)>=0

.compute hmat1(i,j)=1

else

compute hmat1(i,j)=0

end if

end loop

end loop

compute kong=t(hmat)*hmat1

print kong/format "f8.0"

          /title 'Neural network and Fisherian classification'

          /rname=numg/cname=numg

compute prog=diag(kong)

compute diff=abs(dkont-prog)

compute haywan1={dkont,prog,diff}

compute haywan2={"number","prog","diff"}

print haywan1/format "f8.3"

             /title 'Number of objects and accordance of classifications'

             /rname=numg/cname=haywan2

compute tau=1-(csum(diff)&/num)

print tau/format "f8.3"

         /title 'Coefficient of efficacy of neural network'

*-----------------------------------------------------------------

* Indicator matrix in the final classification.

*-----------------------------------------------------------------

compute monkey={"objects", numg}

compute smat={ent,hmat}

*---------------------------------------------------------------------------

* Saving the output files.

*---------------------------------------------------------------------------

save smat /outfile='ftax.sav'

print /title 'Final selector matrix is in the file ftax.sav'

compute ttt=make(ngr,1,1)

loop i=1 to ngr-1

.compute ttt(i+1)=ttt(i)+1

end loop

compute nomvar=s*ttt

compute nomvar={ent,nomvar}

save nomvar /outfile='nomtax.sav'

print /title 'Final nominal variable is in the file nomtax.sav'

*-----------------------------------------------------------------

* End of program and final operations.

*-----------------------------------------------------------------

end matrix

get file='tmp__fds.sav'

restore

!enddefine


Osim ovim programom podaci su analizirani i sa jo{ dva klasi~na taksonomska algoritma, Didayevom metodom pokretnih oblaka i Wardovom metodom hijerahijskog grupisanja. Didayeva varijanta K-means algoritma postigla je koeficijent efikasnosti od 0.941, a Wardova metoda od samo 0.863, tako da je bilo evidentno da se problem detekcije morfolo{kih tipova ne mo`e re{avati primenom metoda koje su uobi~ajene u analizi grupisanja.

3. REZULTATI

Program je, na osnovu scree kriteriuma, procenio da u ovom skupu ima tri taksona. Nakon 16 iteracija dobijeni su rezultati koji su, u vrlo reduciranoj formi, prikazani na slede}im tabelama.

Tabela 1. Aksoni ulaznih neurona na neurone u skrivenom sloju


 g1
g2

TEZINA
1.803
.846

VISINA
-2.983
-1.109

DUZINO
-.364
-.807

BIAKRO
-1.471
.594

BIKRIS
.372
-1.021

NANADL
-.087
-.318

NANALE
-.947
.800

NAPAZU
.392
-1.544

OPNADL
-.098
-2.530

OPPODL
-.525
1.325

OPNATK
2.975
-2.761

OPPOTK
-2.716
3.687

DUZISA
.605
-.342

SIRISA
.564
-.251

NATRBU
-.419
-.024

NAPOTK
-.851
.780

OPGRUD
-.490
1.465

DIRUZG
.138
.396

DILAKT
.572
-.082

DIKOLJ
.439
.119

DUZIST
.138
.054

SISTOP
-1.291
-1.604

DUZIRU
2.391
.857

Tabela 2. Aksoni neurona u skrivenom sloju na neurone u izlaznom sloju


g1
g2
g3

g1
.462
-.854
.238

g2
-.643
-.138
.753


Proces, koji je prepoznatljiv na osnovu konfiguracije signala koji sa ulaznih neurona idu na neurone u skrivenom sloju, a sa ovih na izlazne neurone, razbio je skup entiteta na tri dobro definisana podskupa, ~ije se karakteristike mogu lako prepoznati iz rezultata Fisherove metode diskriminativne analize, tako|e provedene u punom prostoru standardizovanih image varijabli.

Tabela 3. Centroidi finalnih taksona


 g1
g2
g3

TEZINA
-.334
.924
-.423

VISINA
.227
.588
-.706

DUZINO
.395
.471
-.778

BIAKRO
-.193
.777
-.442

BIKRIS
.033
.669
-.579

NANADL
-.230
.774
-.403

NANALE
-.417
.830
-.263

NAPAZU
-.280
.805
-.378

OPNADL
-.687
.876
-.031

OPPODL
-.529
.808
-.133

OPNATK
-.676
1.013
-.154

OPPOTK
-.662
.966
-.129

DUZISA
.248
.404
-.577

SIRISA
-.277
.404
-.053

NATRBU
-.567
.793
-.083

NAPOTK
-.325
.783
-.315

OPGRUD
-.563
.955
-.219

DIRUZG
-.225
.453
-.146

DILAKT
-.214
.638
-.308

DIKOLJ
.138
.674
-.688

DUZIST
.192
.546
-.637

SISTOP
-.033
.571
-.433

DUZIRU
.362
.422
-.705

Tabela 4. Struktura diskriminativnih funkcija


 g1
g2
g3

TEZINA
-.310
.751
-.401

VISINA
.211
.478
-.670

DUZINO
.367
.383
-.738

BIAKRO
-.180
.632
-.419

BIKRIS
.031
.544
-.549

NANADL
-.214
.629
-.382

NANALE
-.387
.675
-.250

NAPAZU
-.260
.655
-.359

OPNADL
-.638
.712
-.030

OPPODL
-.492
.657
-.126

OPNATK
-.628
.823
-.146

OPPOTK
-.615
.785
-.123

DUZISA
.230
.328
-.547

SIRISA
-.258
.328
-.051

NATRBU
-.527
.645
-.078

NAPOTK
-.302
.636
-.299

OPGRUD
-.523
.776
-.207

DIRUZG
-.209
.368
-.138

DILAKT
-.199
.519
-.292

DIKOLJ
.128
.548
-.653

DUZIST
.178
.444
-.604

SISTOP
-.031
.464
-.411

DUZIRU
.336
.343
-.669

Tabela 5. Sklop diskriminativnih funkcija


 g1
g2
g3

TEZINA
-.212
.520
-.280

VISINA
.126
.334
-.447

DUZINO
.227
.270
-.488

BIAKRO
-.126
.438
-.289

BIKRIS
.010
.379
-.370

NANADL
-.148
.436
-.265

NANALE
-.260
.466
-.180

NAPAZU
-.179
.453
-.250

OPNADL
-.421
.490
-.038

OPPODL
-.327
.453
-.099

OPNATK
-.416
.567
-.117

OPPOTK
-.408
.541
-.101

DUZISA
.141
.230
-.363

SIRISA
-.171
.226
-.042

NATRBU
-.349
.444
-.068

NAPOTK
-.205
.440
-.211

OPGRUD
-.349
.535
-.155

DIRUZG
-.140
.254
-.100

DILAKT
-.137
.359
-.203

DIKOLJ
.072
.382
-.438

DUZIST
.106
.310
-.403

SISTOP
-.028
.323
-.279

DUZIRU
.209
.242
-.443

Tabela 6. Korelacije diskriminativnih funkcija


g1
g2
g3

g1
1.000
-.486
-.554

g2
-.486
1.000
-.458

g3
-.554
-.458
1.000

Tabela 7. Kontingencija klasifikacije formirane neuronskom mre`om i klasifikacije formirane Fisherovom metodom diskriminativne analize u standardizovanom image prostoru


g1
g2
g3

g1
261
0
0

g2
2
207
5

g3
0
0
262

Tabela 8. Kontingencija klasifikacije formirane neuronskom mre`om i klasifikacije formirane Fisherovom metodom diskriminativne analize u realnom prostoru


g1
g2
g3

g1
261
0
0

g2
4
204
6

g3
0
0
262

Tabela 9. Broj i postotak ispitanika u taksonima


broj
postotak

g1
261
35.41

g2
214
39.04

g3
262
35.55

Tabela 10. Koeficijenti efikasnosti neuronske mre`e

Image prostor
Realni prostor

.991
.986


Kako se vidi iz ovih rezultata dobijena su tri taksona sa efikasno{}u klasifikacije od 0.991 u image i 0.986 u realnom prostoru. Prvi takson, u kome je bilo 35% ispitanika, identifikovan je kao astenomorfija, drugi, u kome je bilo 29% ispitanika, kao stenomorfija a tre}i, u kome je bilo 36% ispitanika, kao piknomorfija. U prvom taksonu na{li su se ispitanici sa vrlo malom koli~inom mi{i}ne mase, malom koli~inom masnog tikiva i malim transverzalnim, ali prose~nim ili iznadprose~nim longitudinalnim dimenzijama skeleta. U drugom taksonu bili su ispitanici sa visokim vrednostima na svim antropometrijskim varijablama, a naro~ito na merama mi{i}ne mase. U tre}em su taksonu bili ispitanici sa malim i longitudinalnim i transverzalnim dimenzijama skeleta, malom koli~inom masnog tikva i ispodprose~nom koli~inom mi{i}ne mase. Prema tome, ni Sheldonovi ni Kretchmerovi tipovi ne postoje kada se radi o mu{karcima koji se nalaze u stacionarnoj fazi morfolo{kog razvoja, a od pet tipova koje je, analizom razlivenih taksona, prona{la A. Ho{ek, (Ho{ek, 1978) identifikovani su samo tipovi M, K i R.

4. DISKUSIJA


Problem egzistencije i problem identifikacije morfolo{kih tipova svodi se na re{enje nekoliko elementarnih problema koji pripadaju teoriji uzoraka, teoriji merenja i teoriji klasifikacije.


Prvi je izbor dovoljnog broja ispitanika koji su reprezentativni za neku populaciju, homogenu po polu, koja se nalazi u nekoj odre|enoj fazi morfolo{kog razvoja. Ovo zbog toga {to se samo na takvim uzorcima mogu otkriti taksoni koji nisu posledica seksualnog dimorfizma ili razlika u fazama razvoja, i {to samo dovoljno veliki uzorci osiguravaju da se morfolo{ki tipovi, ako stvarno postoje, mogu otkriti bez opasnosti da entiteti sa aberantnim morfolo{kim karakteristikama ne proizvedu pseudo taksone koji ne reprezentiraju nikakv stvarni morfolo{ki tip.


Drugi je izbor antropometrijskih varijabli i tehnika njihovog merenja. Samo takav skup antropometrijskih varijabli koji dovoljno pokriva sve latentne antropometrijske karakteristike mo`e definisati prostor u kome le`e stvarni morfolo{ki tipovi, pod uslovom da su te varijable izmerene sa vrlo malom varijansom pogre{ke, zbog toga {to su sve efikasne metode za detekciju distinktnih taksona osetljive i na prostor u kome tra`e zone zgu{njavanja, i na slu~ajne titraje objekata u tom prostoru.


Tre}i, i verovatno najte`i problem je izbor adekvatnog taksonomskog algoritma. Kao {to je dobro poznato, taksonomski problemi nemaju re{enje u zatvorenoj algebarskoj formi, ve} se svode na neki numeri~ki algoritam kojim se optimizira neka taksonomska funkcija. Za sada nije, i dugo vremena ne}e biti jasno koji je od nekoliko hiljada taksonomskih algoritama koji su do sada predlo`eni najpogodniji za re{avanje problema morfolo{kih tipova. No jedno je ipak jasno na osnovu rezultata analiza provedenih u nekoliko poslednjih godina: Ni jedna od klasi~nih metoda hijerarhijskog grupisanja, i ni jedna od klasi~nih metoda lokalne optimizacije ne proizvode taksone sa dovoljno visokim merama razbijanja i dovoljno visokim koeficijentima efikasnosti (Popovi} i Momirovi}, 2002; Momirovi}, Ho{ek, Popovi} i Boli, 2002). Zbog toga je mo`da primena neuronskih mre`a jedno od mogu}ih re{enja ovog problema.

� Kako su pokazala neka novija istra`ivanja (Momirovi}, Ho{ek, Popovi} and Boli, 2002; Popovi} i Momirovi}, 2002) taksonomske neuronske mre`e su superiornije od klasi~nih taksonomskih algoritama kada se radi o razbijanju nekog skupa objekata na distinktne podskupove bez obzira na to kojim su tipom varijabli ti objekti opisani.





